

for-Schleife

Wiederholt Geometrie für mehrere Werte.
Bereich: [Start : Schritt : Ende].

```
for (x=[0:10:40])  
  translate([x,0,0]) sphere(d=5);
```

Listen durchlaufen

Listen enthalten frei gewählte Werte.
Die Schleifenvariable nimmt jeden Wert an.

```
positionen=[-30,-10,10,30];  
for (x=positionen)  
  translate([x,0,0]) cylinder(h=5,d=4);
```

Verschachtelte Schleifen

Zwei Schleifen erzeugen ein Raster.
Jede X-Position wird mit jeder Y-Position kombiniert.

```
for (x=[-20,20])  
  for (y=[-15,15])  
    translate([x,y,0]) cylinder(h=6,d=5);
```

Bedingungen: if / else

Geometrie abhängig von einem Wert erzeugen.
Bei einzelnen Befehlen sind Klammern optional.

```
senkung=true;  
if (senkung) cylinder(h=2,d1=5,d2=10);  
else cylinder(h=2,d=5);
```

Vergleiche und Logik

Vergleiche: ==, !=, <, <=, >, >=.
Logik: && (und), || (oder), ! (nicht).

```
if (breite>40 && staerke>=3)  
  echo("Abmessungen geeignet");
```

Bedingter Ausdruck

Kurze Auswahl mit ? und :.
Ideal für Vorschau- und Renderqualität.

```
$fn = $preview ? 24 : 96;  
farbe = aktiv ? "green" : "gray";
```

Eigene Module

Module bündeln wiederverwendbare Geometrie.
Der Aufruf erfolgt mit dem Modulnamen.

```
module bolzen() {  
    cylinder(h=12,d=6);  
}  
bolzen();
```

Modulparameter

Parameter machen ein Modul flexibel.
Standardwerte sind optional.

```
module bolzen(d=6,h=12) {  
    cylinder(h=h,d=d);  
}  
bolzen(d=8,h=20);
```

Lokale Werte

Werte innerhalb eines Moduls gelten dort lokal.
Eindeutige Namen erhöhen die Lesbarkeit.

```
module ring(aussen,wand,h) {  
    innen=aussen-2*wand;  
    difference(){ cylinder(h=h,d=aussen);  
                  cylinder(h=h+2,d=innen); }  
}
```

Kinder verwenden

children() setzt übergebene Geometrie ein.
So entstehen eigene Transformationshilfen.

```
module rechts(x=20){  
    translate([x,0,0]) children();  
}  
rechts(30) cube(10);
```

Typische Fehler

Schleife ohne passenden Wertebereich.
== mit = verwechselt; Modul nicht aufgerufen.
Variablen außerhalb ihres Bereichs erwartet.

Mini-Übung

Erzeuge ein 4 × 3 Lochraster.
Lagere die einzelne Bohrung in ein Modul aus.
Blende das Raster über einen Boolean ein oder aus.